
1. [bookmark: _Toc122434485][bookmark: _Toc52895930]Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2020-09_D22wk37 related to the title of this CR.
Contact:	Kalahasti, Narendra
	Narendra.Kalahasti@anritsu.com

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc54186268]Corrections required
2.1 Change 1
	Function name
	f_SIP_BuildSipUri_RX

	Reason for change
	Presently ttcn expects the portField of the requestUri to be omitted while receiving REFER, because function f_SIP_ParseURI always returns portField as omit. But as per section 19.1.2 from RFC 3261, port is an optional field on the requestUri. 

	Summary of change
	Set portField to AnyOrOmit (*) while receiving SipUrl

	TTCN module
	IMS_SIP_Templates.ttcn

	MCC160 Comment
	Accepted in principle: 
The purpose of f_SIP_BuildSipUri_RX is to set the SipUrl according to the given p_URI string (which may or may not have a port number).
 the proposed change manipulates the port number what results in hiding of the issue causing mismatch of the port number.
 The issues causing the problem shall be solve and there is no need to change f_SIP_BuildSipUri_RX.
NOTE: A prose CR is needed for 34.229-1 A.2.11 to correct the Contact URI of the NOTIFY for refer event package to be the SS' Contact URI of the dialog (i.e. the same URI as the Request URI of the corresponding REFER)



Before change:
function f_SIP_BuildSipUri_RX(charstring p_URI,
                                template SemicolonParam_List p_UrlParameters := *,
                                template AmpersandParam_List p_Headers := omit) return template (present) SipUrl
  { /* @sic R5s220953 Further changes by MCC160: p_Port) removed sic@ */
    var template (present) SipUrl v_SipUrl := f_SIP_ParseURI(p_URI);
    
    v_SipUrl.urlParameters := p_UrlParameters;
    v_SipUrl.headers := p_Headers;  /* @sic R5s150382 change 4 sic@ */
    
    return v_SipUrl;
  }

After change:

function f_SIP_BuildSipUri_RX(charstring p_URI,
                                template SemicolonParam_List p_UrlParameters := *,
                                template AmpersandParam_List p_Headers := omit) return template (present) SipUrl
  { /* @sic R5s220953 Further changes by MCC160: p_Port) removed sic@ */
    var template (present) SipUrl v_SipUrl := f_SIP_ParseURI(p_URI);
    
    if (ischosen(v_SipUrl.components.sip)) {
      v_SipUrl.components.sip.hostPort.portField := *;
    }
	v_SipUrl.urlParameters := p_UrlParameters;
    v_SipUrl.headers := p_Headers;  /* @sic R5s150382 change 4 sic@ */
    
    return v_SipUrl;
  }     

MCC160 Implementation
	[bookmark: _Hlk120009264]  altstep a_IMS_InviteUserToConference(inout InviteUserToConference_StateInfo_Type p_StateInfo,
                                       charstring p_FinalConferenceUri,
                                       template (omit) IMS_PTC_DialogIndex_Type p_ActiveSessionToBeReplaced,
                                       template (omit) SUBSCRIBE_Request p_SubscribeRequest,
                                       CharStringList_Type p_MediaTypeList,
                                       integer p_ConferenceEventPackage_Version := 1) runs on IMS_PTC
  { /* @sic R5-153765, R5s150909: new parameter p_ActiveSessionToBeReplaced sic@ */
    /* @sic R5s160055: p_SubscriptionState_Step5 removed (v_SubscriptionState_Step5 is used instead) sic@ */
    var IMS_DATA_REQ v_IMS_REFER_REQ_ByRef;
    var InviteUserToConference_State_Type v_State := p_StateInfo.State;
    var MessageHeader v_MessageHeader_Refer;
    var template (value) MessageHeader v_MessageHeader_Response;
    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;
    var SUBSCRIBE_Request v_SubscribeRequest;
    var template (value) SipUrl v_ContactUrl;
    var InternetProtocol_Type v_ProtocolForRequestInDL := f_IMS_PTC_ImsInfo_GetTransportProtocol();
    var template (value) SubscriptionState v_SubscriptionState_Step5 := cs_SubscriptionState("terminated", { cs_GenericParam("reason", "noresource") } );  /* @sic R5s141343 change 3.3 and 4; R5s160055 sic@ */

    // Step 1. UE sends REFER to SS
    [v_State == waitForREFER] a_IMS_REFER_ReceiveRequest(v_IMS_REFER_REQ_ByRef, p_FinalConferenceUri, p_StateInfo.CalleeUri, p_ActiveSessionToBeReplaced) {
      v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_REFER_REQ_ByRef.RoutingInfo);
      v_MessageHeader_Refer := v_IMS_REFER_REQ_ByRef.Request.Refer.msgHeader;

      // Step 2. The SS responds with 202 Accepted
      v_MessageHeader_Response := f_IMS_Accepted_202_MessageHeaderTX(v_MessageHeader_Refer);
      IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine202, v_MessageHeader_Response)));

      // Step 3. The SS sends initial NOTIFY for the implicit subscription created by the REFER request
      p_StateInfo.ContactUriUE := f_MessageHeader_GetContactSipUrl(v_MessageHeader_Refer);      // UE's contact to be use in the request line of the NOTIFY messages
      v_ContactUrl := f_SIP_BuildSipUri_TX(p_FinalConferenceUri, f_IMS_PTC_GetPort_ps());       // contact uri of the conference focus

      p_StateInfo.NotifyHeader := valueof(f_IMS_NotifyRequest_MessageHeader_A_2_11(-, v_ContactUrl, cs_SubscriptionState_Active("300")));
      IMS_Client.send(cas_IMS_Notify_Request(cs_IMS_RoutingInfo(v_ProtocolForRequestInDL),
                                             cs_NOTIFY_Request(p_StateInfo.ContactUriUE, p_StateInfo.NotifyHeader, cs_MessageBody_SipFrag("SIP/2.0 100 Trying"))));
      p_StateInfo.State := waitFor200OK_Step4;
    }
    // Step 4. UE responds with 200 OK
    [v_State == waitFor200OK_Step4] a_IMS_ReceiveResponse200OK(p_StateInfo.NotifyHeader, v_ProtocolForRequestInDL) {
      // Step 5. The SS sends a NOTIFY related to REFER request to confirm that the invited user was able to join the conference
      v_ContactUrl := f_SIP_BuildSipUri_TX(p_FinalConferenceUri, f_IMS_PTC_GetPort_ps());       // contact uri of the conference focus
      p_StateInfo.NotifyHeader := valueof(f_IMS_NotifyRequest_MessageHeader_A_2_11(-, v_ContactUrl, v_SubscriptionState_Step5));  // @sic R5s221176: v_ContactUrl instead of p_StateInfo.ContactUriUE sic@
      IMS_Client.send(cas_IMS_Notify_Request(cs_IMS_RoutingInfo(v_ProtocolForRequestInDL),
                                             cs_NOTIFY_Request(p_StateInfo.ContactUriUE, p_StateInfo.NotifyHeader, cs_MessageBody_SipFrag("SIP/2.0 200 OK"))));
      p_StateInfo.State := waitFor200OK_Step6;
    }
    // Step 6. UE responds with 200 OK
    [v_State == waitFor200OK_Step6] a_IMS_ReceiveResponse200OK(p_StateInfo.NotifyHeader, v_ProtocolForRequestInDL) {
      if (ispresent(p_SubscribeRequest)) {
        v_SubscribeRequest := valueof(p_SubscribeRequest);

        // Step 7. The SS sends a NOTIFY for conference event package to inform that the invited user was able to join the conference
        p_StateInfo.NotifyHeader := valueof(f_IMS_ConferenceEventPackage_SendNotifySubscribe(v_SubscribeRequest,
                                                                                             v_ProtocolForRequestInDL,
                                                                                             p_FinalConferenceUri,
                                                                                             p_StateInfo.CalleeUri,
                                                                                             p_StateInfo.CalleeContactUri,
                                                                                             p_MediaTypeList,
                                                                                             p_ConferenceEventPackage_Version,
                                                                                             partial));
        p_StateInfo.State := waitFor200OK_Step8;
      }
      else {
        p_StateInfo.State := finished;
      }
    }
    // Step 8. UE responds with 200 OK
    [v_State == waitFor200OK_Step8] a_IMS_ReceiveResponse200OK(p_StateInfo.NotifyHeader, v_ProtocolForRequestInDL) {
      p_StateInfo.State := finished;
    }
  }



Further Changes
	  function f_TC_15_25_IMS1() runs on IMS_PTC
  { // MO Explicit Communication Transfer - Consultative Call Transfer
    var charstring v_CalleeUri_Dialog1 := px_IMS_CalleeUri;
    var charstring v_CalleeUri_Dialog2 := px_IMS_CalleeUri2;
    var ThreeWaySession_SessionInfo_Type v_SessionInfo_Dialog1;
    var ThreeWaySession_SessionInfo_Type v_SessionInfo_Dialog2;
    var charstring v_CalleeContactUri_Dialog1 := px_IMS_CalleeContactUri;
    var template (value) SipUrl v_ContactUrl_1;
    var SipUrl v_UEContactUrl;
    var template (value)  SDP_Message v_SDP_LatestMessageBySS;
    var boolean v_DontIncrementSessVers;
    
    f_IMS_CC_Preamble(IPCAN_SpeechCall, IMS_REGISTERED);

    // Setup MO speech call
    f_IMS_CC_StartCall(IPCAN_MO_SpeechCall, v_CalleeUri_Dialog1);
    
    // Call establishment (Annex C.21 / Dialog-1) in preamble
    v_SessionInfo_Dialog1.InviteRequestWithSdp := f_IMS_MOCallSetup_Step2(IPCAN_MO_SpeechCall, v_CalleeUri_Dialog1);
    v_UEContactUrl := f_Addr_Union_GetSipUrl(v_SessionInfo_Dialog1.InviteRequestWithSdp.Invite.msgHeader.contact.contactBody.contactAddresses[0].addressField);
    
    v_SessionInfo_Dialog1.MessageHeader_PrevResponse := f_IMS_MOCallSetup_AnnexC21_Step3_4(v_SessionInfo_Dialog1.InviteRequestWithSdp);
    f_IMS_CC_TriggerDedicatedBearerActivation(IPCAN_MO_SpeechCall);
    v_SessionInfo_Dialog1.LatestSdpOffer := f_IMS_MOCallSetup_AnnexC21_Steps5_13(v_SessionInfo_Dialog1.InviteRequestWithSdp, v_SessionInfo_Dialog1.MessageHeader_PrevResponse);
    v_SessionInfo_Dialog1.ContactUrl := f_MessageHeader_GetContactSipUrl(v_SessionInfo_Dialog1.MessageHeader_PrevResponse);
    v_DontIncrementSessVers := true;  // @sic R5s211703: session version shall not be incremented sic@
    v_SDP_LatestMessageBySS := f_IMS_BuildSDP_UpdatedAnswer_TX(v_SessionInfo_Dialog1.LatestSdpOffer, v_DontIncrementSessVers);

    v_ContactUrl_1 := f_SIP_BuildSipUri_TX(v_CalleeContactUri_Dialog1, f_IMS_PTC_GetPort_ps());
        
    f_IMS_TestBody_Set(true);
    
    // Initiate Explicit Communication Transfer (ECT) - 3GPP TS 24.629
    f_UT_IMS_ExplicitCallTransfer(MMI, v_CalleeUri_Dialog2);
    
    
    // Steps 1-4
    /* Steps defined in Annex C.8 to hold the call.
     * Note: UE starts automatically the call hold procedure
     */
    f_IMS_MOCallHoldResume(HOLD, v_SessionInfo_Dialog1.InviteRequestWithSdp.Invite, v_SessionInfo_Dialog1.MessageHeader_PrevResponse, v_SessionInfo_Dialog1.LatestSdpOffer);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "");
 

    // Steps 5-16 (Annex C.21 / Dialog-2)
    f_IMS_PTC_ImsInfo_ActivateDialog(secondDialog);
    f_IMS_PTC_ImsInfo_DialogSetSessIdTX("11111112");      // @sic R5-220212 sic@

    //  Note: UE starts automatically the call establishment procedure
    v_SessionInfo_Dialog2.InviteRequestWithSdp := f_IMS_MOCallSetup_Step2(IPCAN_MO_SpeechCall, v_CalleeUri_Dialog2);
    v_SessionInfo_Dialog2.MessageHeader_PrevResponse := f_IMS_MOCallSetup_AnnexC21_Step3_4(v_SessionInfo_Dialog2.InviteRequestWithSdp, px_IMS_CalleeContactUri2);
    v_SessionInfo_Dialog2.LatestSdpOffer := f_IMS_MOCallSetup_AnnexC21_Steps5_13(v_SessionInfo_Dialog2.InviteRequestWithSdp, v_SessionInfo_Dialog2.MessageHeader_PrevResponse, px_IMS_CalleeContactUri2);
    v_SessionInfo_Dialog2.ContactUrl := f_MessageHeader_GetContactSipUrl(v_SessionInfo_Dialog2.MessageHeader_PrevResponse);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "");
    
    // Steps 17-20
    /* Steps defined in Annex C.8 to hold the call (on second call.
     * Note: UE starts automatically the call hold procedure
     */
    f_IMS_MOCallHoldResume(HOLD, v_SessionInfo_Dialog2.InviteRequestWithSdp.Invite, v_SessionInfo_Dialog2.MessageHeader_PrevResponse, v_SessionInfo_Dialog2.LatestSdpOffer, px_IMS_CalleeContactUri2);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "");
    
    f_IMS_PTC_ImsInfo_ActivateDialog(firstDialog);

    // Steps 21-22: Receive REFER, send 200 OK for REFER
    f_IMS_CallTransfer_Refer(v_CalleeContactUri_Dialog1, v_CalleeUri_Dialog2, secondDialog);                            // @sic R5-211446: f_IMS_CallTransfer_Refer sic@
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "");

    // Steps 23-24: Send NOTIFY, receive 200 OK for NOTIFY
    f_IMS_CallTransfer_Notify_Active(v_ContactUrl_1, v_UEContactUrl);                                                   // @sic R5-211446: f_IMS_CallTransfer_Notify_Active sic@
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "");

    // Steps 25-28
    // Steps defined in Annex C.9 to hold the call. The same messages as in Annex C.9 Steps 1-4 are used.
    f_IMS_MTCallHoldResume(HOLDINACTIVE, v_UEContactUrl, v_SDP_LatestMessageBySS, v_ContactUrl_1);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "");
    
    // Steps 29-30 BYE/200 OK on Session-2 (UE <-> Target)
    f_IMS_PTC_ImsInfo_ActivateDialog(secondDialog);
    f_IMS_CallReleaseMT(v_UEContactUrl);
    
    // Steps 31-32
    // The SS, simulating the transferee, sends a NOTIFY to confirm that the call transfer has been completed
    f_IMS_PTC_ImsInfo_ActivateDialog(firstDialog);
    f_IMS_CallTransfer_Notify_200OK_Terminated(v_ContactUrl_1, v_UEContactUrl);                                         // @sic R5-211446: f_IMS_CallTransfer_Notify_Active sic@

    // Steps 33 to 34
    // Optional: UE may send BYE request to release call with transferee UE
    f_IMS_OptionalCallReleaseMO(v_SessionInfo_Dialog1.InviteRequestWithSdp.Invite, v_SessionInfo_Dialog1.ContactUrl);   // @sic R5-211446: f_IMS_OptionalCallReleaseMO sic@
    
    f_IMS_TestBody_Set(false);

    f_IMS_CC_Postamble(IPCAN_MO_SpeechCall);
  }



2.2 Change 2
	Function name
	f_IMS_ThreeWaySession_SubscribeInvite

	Reason for change
	Due to incorrect condition being used for the do-while loop, in the case when UE sends SUBSCRIBE message after termination of both dialogs, the ttcn may unexpectedly exit the loop even before receiving the SUBSCRIBE message.

	Summary of change
	Introduced not for the condition which checks for the value of variable v_SubscribeRequest

	TTCN module
	IMS_Procedures_Conferencing.ttcn

	MCC160 Comment
	Accepted in principle.



Before change:
function f_IMS_ThreeWaySession_SubscribeInvite(ThreeWaySession_SessionInfo_Type p_SessionInfo_Dialog1,
                                                 ThreeWaySession_SessionInfo_Type p_SessionInfo_Dialog2,
                                                 InviteUserToConference_StateInfo_Type p_Dialog1_StateInfo,
                                                 InviteUserToConference_StateInfo_Type p_Dialog2_StateInfo,
                                                 charstring p_FinalConferenceUri,
                                                 CharStringList_Type p_MediaTypeList) runs on IMS_PTC return template (omit) SUBSCRIBE_Request
  { /* @sic R5-214877: p_MediaTypeList sic@ */
    var ConferenceEventPackage_SubscribeInfo_Type v_ConferenceEventPackage_SubscribeInfo_ByRef := valueof(cs_ConferenceEventPackage_SubscribeInfo_Init);
    var InviteUserToConference_StateInfo_Type v_Dialog1_StateInfo_ByRef := p_Dialog1_StateInfo;
    var InviteUserToConference_StateInfo_Type v_Dialog2_StateInfo_ByRef := p_Dialog2_StateInfo;
    
    var template (present) MessageHeader v_MessageHeader_Bye_Dialog1 := f_IMS_ByeRequest_MessageHeaderRX(p_SessionInfo_Dialog1.InviteRequestWithSdp.Invite, -, firstDialog);
    var template (present) MessageHeader v_MessageHeader_Bye_Dialog2 := f_IMS_ByeRequest_MessageHeaderRX(p_SessionInfo_Dialog2.InviteRequestWithSdp.Invite, -, secondDialog);
    var template (omit) SUBSCRIBE_Request v_SubscribeRequest := omit;

    var boolean v_WaitForBYE_Dialog1 := false;
    var boolean v_WaitForBYE_Dialog2 := false;
    var boolean v_Dialog1_Terminated := false;
    var boolean v_Dialog2_Terminated := false;
    var boolean v_Continue;

    var integer v_ConferenceEventPackage_Version := 1;   // version used in notifications sent to the UE to inform about remote user has joined the conference

    do {
      alt {
        // 34.229-1 C.10/38 Step 10 - 13, 34.229-5 A.19/25 Step 10 - 13         (UE subscribes the conference event)
        [v_ConferenceEventPackage_SubscribeInfo_ByRef.State != subscribed] a_ConferenceEventPackage_Subscribe(v_ConferenceEventPackage_SubscribeInfo_ByRef, p_FinalConferenceUri)
          {
            v_SubscribeRequest := v_ConferenceEventPackage_SubscribeInfo_ByRef.SubscribeRequest;
          }
        // 34.229-1 C.19/37, 34.229-5 A.20/26                                   (Invite user A)
        [v_Dialog1_StateInfo_ByRef.State != finished] a_IMS_InviteUserToConference(v_Dialog1_StateInfo_ByRef,
                                                                                   p_FinalConferenceUri,
                                                                                   firstDialog,       // @sic R5s150909: dialog index instead of boolean sic@
                                                                                   v_SubscribeRequest,
                                                                                   p_MediaTypeList,   // @sic R5-214877: p_MediaTypeList instead of tsc_MediaTypeList_Audio sic@
                                                                                   v_ConferenceEventPackage_Version)
          {
            v_WaitForBYE_Dialog1 := fl_IMS_ThreeWaySession_StateInfo_WaitForBYE(v_Dialog1_StateInfo_ByRef) and not v_Dialog1_Terminated;   /* determine whether to wait for BYE for dialog 1; @sic R5s150910: v_Dialog1_Terminated sic@ */
            if (v_Dialog1_StateInfo_ByRef.State == waitFor200OK_Step8) {                              /* => when then UE has subscribed to the conference event package it gets notify that user 1 has joined the conference ... */
              v_ConferenceEventPackage_Version := v_ConferenceEventPackage_Version + 1;               /* ... and the version needs to incremented for next notification */
            }
          }
        // 34.229-1 C.19/37, 34.229-5 A.20/26                                   (Invite user A)
        [v_Dialog2_StateInfo_ByRef.State != finished] a_IMS_InviteUserToConference(v_Dialog2_StateInfo_ByRef,
                                                                                   p_FinalConferenceUri,
                                                                                   secondDialog,      // @sic R5s150909: dialog index instead of boolean sic@
                                                                                   v_SubscribeRequest,
                                                                                   p_MediaTypeList,   // @sic R5-214877: p_MediaTypeList instead of tsc_MediaTypeList_Audio sic@
                                                                                   v_ConferenceEventPackage_Version)
          {
            v_WaitForBYE_Dialog2 := fl_IMS_ThreeWaySession_StateInfo_WaitForBYE(v_Dialog2_StateInfo_ByRef) and not v_Dialog2_Terminated;   /* determine whether to wait for BYE for dialog 2; @sic R5s150910: v_Dialog2_Terminated sic@ */
            if (v_Dialog2_StateInfo_ByRef.State == waitFor200OK_Step8) {                              /* => when then UE has subscribed to the conference event package it gets notify that user 2 has joined the conference ... */
              v_ConferenceEventPackage_Version := v_ConferenceEventPackage_Version + 1;               /* ... and the version needs to incremented for next notification */
            }
          }
        //--------------------------------------------
        // DIALOG 1
        //--------------------------------------------
        // Terminate dialog 1
        [v_WaitForBYE_Dialog1] a_IMS_ReceiveByeSendOK(firstDialog, p_SessionInfo_Dialog1.ContactUrl, v_MessageHeader_Bye_Dialog1)
          {
            v_WaitForBYE_Dialog1 := false;
            v_Dialog1_Terminated := true;   /* @sic R5s150910 sic@ */
          }
        
        //--------------------------------------------
        // DIALOG 2
        //--------------------------------------------
        // Terminate dialog 2
        [v_WaitForBYE_Dialog2] a_IMS_ReceiveByeSendOK(secondDialog, p_SessionInfo_Dialog2.ContactUrl, v_MessageHeader_Bye_Dialog2)
          {
            v_WaitForBYE_Dialog2 := false;
            v_Dialog2_Terminated := true;   /* @sic R5s150910 sic@ */
          }
      }
      v_Continue := v_WaitForBYE_Dialog1 or v_WaitForBYE_Dialog2
        or (v_Dialog1_StateInfo_ByRef.State != finished) or (v_Dialog2_StateInfo_ByRef.State != finished)
        or (isvalue(v_SubscribeRequest) and (v_ConferenceEventPackage_SubscribeInfo_ByRef.State != subscribed));
      
    } while (v_Continue);

    return v_SubscribeRequest;
  }


After change:

function f_IMS_ThreeWaySession_SubscribeInvite(ThreeWaySession_SessionInfo_Type p_SessionInfo_Dialog1,
                                                 ThreeWaySession_SessionInfo_Type p_SessionInfo_Dialog2,
                                                 InviteUserToConference_StateInfo_Type p_Dialog1_StateInfo,
                                                 InviteUserToConference_StateInfo_Type p_Dialog2_StateInfo,
                                                 charstring p_FinalConferenceUri,
                                                 CharStringList_Type p_MediaTypeList) runs on IMS_PTC return template (omit) SUBSCRIBE_Request
  { /* @sic R5-214877: p_MediaTypeList sic@ */
    var ConferenceEventPackage_SubscribeInfo_Type v_ConferenceEventPackage_SubscribeInfo_ByRef := valueof(cs_ConferenceEventPackage_SubscribeInfo_Init);
    var InviteUserToConference_StateInfo_Type v_Dialog1_StateInfo_ByRef := p_Dialog1_StateInfo;
    var InviteUserToConference_StateInfo_Type v_Dialog2_StateInfo_ByRef := p_Dialog2_StateInfo;
    
    var template (present) MessageHeader v_MessageHeader_Bye_Dialog1 := f_IMS_ByeRequest_MessageHeaderRX(p_SessionInfo_Dialog1.InviteRequestWithSdp.Invite, -, firstDialog);
    var template (present) MessageHeader v_MessageHeader_Bye_Dialog2 := f_IMS_ByeRequest_MessageHeaderRX(p_SessionInfo_Dialog2.InviteRequestWithSdp.Invite, -, secondDialog);
    var template (omit) SUBSCRIBE_Request v_SubscribeRequest := omit;

    var boolean v_WaitForBYE_Dialog1 := false;
    var boolean v_WaitForBYE_Dialog2 := false;
    var boolean v_Dialog1_Terminated := false;
    var boolean v_Dialog2_Terminated := false;
    var boolean v_Continue;

    var integer v_ConferenceEventPackage_Version := 1;   // version used in notifications sent to the UE to inform about remote user has joined the conference

    do {
      alt {
        // 34.229-1 C.10/38 Step 10 - 13, 34.229-5 A.19/25 Step 10 - 13         (UE subscribes the conference event)
        [v_ConferenceEventPackage_SubscribeInfo_ByRef.State != subscribed] a_ConferenceEventPackage_Subscribe(v_ConferenceEventPackage_SubscribeInfo_ByRef, p_FinalConferenceUri)
          {
            v_SubscribeRequest := v_ConferenceEventPackage_SubscribeInfo_ByRef.SubscribeRequest;
          }
        // 34.229-1 C.19/37, 34.229-5 A.20/26                                   (Invite user A)
        [v_Dialog1_StateInfo_ByRef.State != finished] a_IMS_InviteUserToConference(v_Dialog1_StateInfo_ByRef,
                                                                                   p_FinalConferenceUri,
                                                                                   firstDialog,       // @sic R5s150909: dialog index instead of boolean sic@
                                                                                   v_SubscribeRequest,
                                                                                   p_MediaTypeList,   // @sic R5-214877: p_MediaTypeList instead of tsc_MediaTypeList_Audio sic@
                                                                                   v_ConferenceEventPackage_Version)
          {
            v_WaitForBYE_Dialog1 := fl_IMS_ThreeWaySession_StateInfo_WaitForBYE(v_Dialog1_StateInfo_ByRef) and not v_Dialog1_Terminated;   /* determine whether to wait for BYE for dialog 1; @sic R5s150910: v_Dialog1_Terminated sic@ */
            if (v_Dialog1_StateInfo_ByRef.State == waitFor200OK_Step8) {                              /* => when then UE has subscribed to the conference event package it gets notify that user 1 has joined the conference ... */
              v_ConferenceEventPackage_Version := v_ConferenceEventPackage_Version + 1;               /* ... and the version needs to incremented for next notification */
            }
          }
        // 34.229-1 C.19/37, 34.229-5 A.20/26                                   (Invite user A)
        [v_Dialog2_StateInfo_ByRef.State != finished] a_IMS_InviteUserToConference(v_Dialog2_StateInfo_ByRef,
                                                                                   p_FinalConferenceUri,
                                                                                   secondDialog,      // @sic R5s150909: dialog index instead of boolean sic@
                                                                                   v_SubscribeRequest,
                                                                                   p_MediaTypeList,   // @sic R5-214877: p_MediaTypeList instead of tsc_MediaTypeList_Audio sic@
                                                                                   v_ConferenceEventPackage_Version)
          {
            v_WaitForBYE_Dialog2 := fl_IMS_ThreeWaySession_StateInfo_WaitForBYE(v_Dialog2_StateInfo_ByRef) and not v_Dialog2_Terminated;   /* determine whether to wait for BYE for dialog 2; @sic R5s150910: v_Dialog2_Terminated sic@ */
            if (v_Dialog2_StateInfo_ByRef.State == waitFor200OK_Step8) {                              /* => when then UE has subscribed to the conference event package it gets notify that user 2 has joined the conference ... */
              v_ConferenceEventPackage_Version := v_ConferenceEventPackage_Version + 1;               /* ... and the version needs to incremented for next notification */
            }
          }
        //--------------------------------------------
        // DIALOG 1
        //--------------------------------------------
        // Terminate dialog 1
        [v_WaitForBYE_Dialog1] a_IMS_ReceiveByeSendOK(firstDialog, p_SessionInfo_Dialog1.ContactUrl, v_MessageHeader_Bye_Dialog1)
          {
            v_WaitForBYE_Dialog1 := false;
            v_Dialog1_Terminated := true;   /* @sic R5s150910 sic@ */
          }
        
        //--------------------------------------------
        // DIALOG 2
        //--------------------------------------------
        // Terminate dialog 2
        [v_WaitForBYE_Dialog2] a_IMS_ReceiveByeSendOK(secondDialog, p_SessionInfo_Dialog2.ContactUrl, v_MessageHeader_Bye_Dialog2)
          {
            v_WaitForBYE_Dialog2 := false;
            v_Dialog2_Terminated := true;   /* @sic R5s150910 sic@ */
          }
      }
      v_Continue := v_WaitForBYE_Dialog1 or v_WaitForBYE_Dialog2
        or (v_Dialog1_StateInfo_ByRef.State != finished) or (v_Dialog2_StateInfo_ByRef.State != finished)
        or (not isvalue(v_SubscribeRequest) and (v_ConferenceEventPackage_SubscribeInfo_ByRef.State != subscribed));
      
    } while (v_Continue);

    return v_SubscribeRequest;
  }


MCC160 Implementation
	  function f_IMS_ThreeWaySession_SubscribeInvite(ThreeWaySession_SessionInfo_Type p_SessionInfo_Dialog1,
                                                 ThreeWaySession_SessionInfo_Type p_SessionInfo_Dialog2,
                                                 InviteUserToConference_StateInfo_Type p_Dialog1_StateInfo,
                                                 InviteUserToConference_StateInfo_Type p_Dialog2_StateInfo,
                                                 charstring p_FinalConferenceUri,
                                                 CharStringList_Type p_MediaTypeList) runs on IMS_PTC return template (omit) SUBSCRIBE_Request
  { /* @sic R5-214877: p_MediaTypeList sic@ */
    var ConferenceEventPackage_SubscribeInfo_Type v_ConferenceEventPackage_SubscribeInfo_ByRef := valueof(cs_ConferenceEventPackage_SubscribeInfo_Init);
    var InviteUserToConference_StateInfo_Type v_Dialog1_StateInfo_ByRef := p_Dialog1_StateInfo;
    var InviteUserToConference_StateInfo_Type v_Dialog2_StateInfo_ByRef := p_Dialog2_StateInfo;
    
    var template (present) MessageHeader v_MessageHeader_Bye_Dialog1 := f_IMS_ByeRequest_MessageHeaderRX(p_SessionInfo_Dialog1.InviteRequestWithSdp.Invite, -, firstDialog);
    var template (present) MessageHeader v_MessageHeader_Bye_Dialog2 := f_IMS_ByeRequest_MessageHeaderRX(p_SessionInfo_Dialog2.InviteRequestWithSdp.Invite, -, secondDialog);
    var template (omit) SUBSCRIBE_Request v_SubscribeRequest := omit;

    var boolean v_WaitForBYE_Dialog1 := false;
    var boolean v_WaitForBYE_Dialog2 := false;
    var boolean v_Dialog1_Terminated := false;
    var boolean v_Dialog2_Terminated := false;
    var boolean v_Continue;

    var integer v_ConferenceEventPackage_Version := 1;   // version used in notifications sent to the UE to inform about remote user has joined the conference

    timer t_WaitForSubscribe := tsc_WaitForConferenceSubscription;

    t_WaitForSubscribe.start;

    do {
      alt {
        // 34.229-1 C.10/38 Step 10 - 13, 34.229-5 A.19/25 Step 10 - 13         (UE subscribes the conference event)
        [v_ConferenceEventPackage_SubscribeInfo_ByRef.State != subscribed] a_ConferenceEventPackage_Subscribe(v_ConferenceEventPackage_SubscribeInfo_ByRef, p_FinalConferenceUri)
          {
            v_SubscribeRequest := v_ConferenceEventPackage_SubscribeInfo_ByRef.SubscribeRequest;
            t_WaitForSubscribe.stop;
          }
        // 34.229-1 C.19/37, 34.229-5 A.20/26                                   (Invite user A)
        [v_Dialog1_StateInfo_ByRef.State != finished] a_IMS_InviteUserToConference(v_Dialog1_StateInfo_ByRef,
                                                                                   p_FinalConferenceUri,
                                                                                   firstDialog,       // @sic R5s150909: dialog index instead of boolean sic@
                                                                                   v_SubscribeRequest,
                                                                                   p_MediaTypeList,   // @sic R5-214877: p_MediaTypeList instead of tsc_MediaTypeList_Audio sic@
                                                                                   v_ConferenceEventPackage_Version)
          {
            v_WaitForBYE_Dialog1 := fl_IMS_ThreeWaySession_StateInfo_WaitForBYE(v_Dialog1_StateInfo_ByRef) and not v_Dialog1_Terminated;   /* determine whether to wait for BYE for dialog 1; @sic R5s150910: v_Dialog1_Terminated sic@ */
            if (v_Dialog1_StateInfo_ByRef.State == waitFor200OK_Step8) {                              /* => when then UE has subscribed to the conference event package it gets notify that user 1 has joined the conference ... */
              v_ConferenceEventPackage_Version := v_ConferenceEventPackage_Version + 1;               /* ... and the version needs to incremented for next notification */
            }
          }
        // 34.229-1 C.19/37, 34.229-5 A.20/26                                   (Invite user A)
        [v_Dialog2_StateInfo_ByRef.State != finished] a_IMS_InviteUserToConference(v_Dialog2_StateInfo_ByRef,
                                                                                   p_FinalConferenceUri,
                                                                                   secondDialog,      // @sic R5s150909: dialog index instead of boolean sic@
                                                                                   v_SubscribeRequest,
                                                                                   p_MediaTypeList,   // @sic R5-214877: p_MediaTypeList instead of tsc_MediaTypeList_Audio sic@
                                                                                   v_ConferenceEventPackage_Version)
          {
            v_WaitForBYE_Dialog2 := fl_IMS_ThreeWaySession_StateInfo_WaitForBYE(v_Dialog2_StateInfo_ByRef) and not v_Dialog2_Terminated;   /* determine whether to wait for BYE for dialog 2; @sic R5s150910: v_Dialog2_Terminated sic@ */
            if (v_Dialog2_StateInfo_ByRef.State == waitFor200OK_Step8) {                              /* => when then UE has subscribed to the conference event package it gets notify that user 2 has joined the conference ... */
              v_ConferenceEventPackage_Version := v_ConferenceEventPackage_Version + 1;               /* ... and the version needs to incremented for next notification */
            }
          }
        //--------------------------------------------
        // DIALOG 1
        //--------------------------------------------
        // Terminate dialog 1
        [v_WaitForBYE_Dialog1] a_IMS_ReceiveByeSendOK(firstDialog, p_SessionInfo_Dialog1.ContactUrl, v_MessageHeader_Bye_Dialog1)
          {
            v_WaitForBYE_Dialog1 := false;
            v_Dialog1_Terminated := true;   /* @sic R5s150910 sic@ */
          }
        
        //--------------------------------------------
        // DIALOG 2
        //--------------------------------------------
        // Terminate dialog 2
        [v_WaitForBYE_Dialog2] a_IMS_ReceiveByeSendOK(secondDialog, p_SessionInfo_Dialog2.ContactUrl, v_MessageHeader_Bye_Dialog2)
          {
            v_WaitForBYE_Dialog2 := false;
            v_Dialog2_Terminated := true;   /* @sic R5s150910 sic@ */
          }
      }
      v_Continue := v_WaitForBYE_Dialog1 or v_WaitForBYE_Dialog2
        or (v_Dialog1_StateInfo_ByRef.State != finished) or (v_Dialog2_StateInfo_ByRef.State != finished)
        or (isvalue(v_SubscribeRequest) and (v_ConferenceEventPackage_SubscribeInfo_ByRef.State != subscribed))
        or t_WaitForSubscribe.running;
      
    } while (v_Continue);

    return v_SubscribeRequest;
  }




